DECUS
PROGRAM LIBRARY

DECUS NO.	8-537
TITLE	A FLEXIBLE DATA BUFFER DISPLAY ROUTINE FOR LAB-8 SYSTEMS
AUTHOR	A. J. Swan
COMPANY	Agricultural Research Council's Poultry Research Center Edinburgh, Scotland
DATE	July 6, 1973
SOURCE LANGUAGE	PAL III

A FLEXIBLE DATA BUFFER DISPLAY ROUTINE FOR LAB-8 SYSTEMS

DECUS Program Library Write-up

DECUS NO. 8-637

ABSTRACT

This set of five subroutines may be called to display the contents of a data buffer on the oscilloscope of an AX-08 (LAB-8) system. Software control of format as either a point or a histogram display, vertical scaling and placement in the two axes of the display is provided.

REQUIREMENTS

Storage - these subroutines occupy 162(8) storage locations and may be placed anywhere in memory on a single page.

Equipment - basic LAB-8 system with oscilloscope display.

USAGE

Loading - The subroutines are supplied as a PAL symbolic tape with no origin setting, but terminated with a PAUSE directive, which may be assembled to any convenient location.

Calling Sequence - The display subroutine is called by an effective JMS PLOTER with the relevant parameters (detailed elsewhere) set by a user provided subroutine. As a call to the display generation subroutine will only display the data buffer a single time, multiple calls are required to provide the refreshing necessary for a visual display. Certain parameters are destroyed by the routine and must be reset between successive calls. It is expected that the reset routine will provide the facilities for changing display format and as this is likely to be program specific no routine is supplied. The vertical scale controlling subroutines are called by an effective JMS VSFMUL to increase the vertical scale or an effective JMS VSFDIV to decrease the vertical scale.

RESTRICTIONS

Hardware - The standard AX-08 display restriction applies, namely no display is possible during the interval between an ADCV and subsequent RADC instruction. The normal display problem of "wrap-around" on both axes also holds, therefore data must be kept in the relevant range for the display.

DESCRIPTION

The subroutine PLOTER generates a display according to the parameters set in various locations when it is called as detailed below.

BUFFER is set to point to the first location in the data buffer to be displayed and NOPTS is set to minus the number of points in the buffer. XAXIS and BASELN set the position of the display in the X and Y axes. XAXIS should be set to the initial point on the screen where the display is to start. BASELN is added to every data point, after any scaling has been carried out, and is especially useful when generating histograms as the effective zero for the display may be moved to the bottom of the display screen (-377 octal). XSTEP controls the distance between successive points on the display face and obviously should be selected to prevent "wrap around" on the X-axis of the display. RBIN controls the form of the data display. If it is set to -1 then the data is displayed as individual points whereas if it is set to -2 or less then a histogram display will be generated with each bin being that number of display points wide (each point being XSTEP apart). It should be noted that if the first or last values in the data buffer are non-zero then the routine will not join the leading or trailing edges respectively to the effective zero as defined by BASELN. Time is saved by not displaying null data because the routine automatically generates a baseline across the display from XAXIS to XEND at effective zero by calling the routine BASLIN on entry. It is possible to ensure that no baseline is drawn by setting XAXIS to the same value as XEND before the call to PLOTER.

SCALE1 and SCALE2 provide a capability for vertical scaling of the display data (without buffer modification) as defined by Table 1. The routines VSFMUL, VSFDIV and VSETUP provide a means of achievding each of these scaling factors under program

control. An effective JMS VSETUP with a cleared accumulator returns the vertical scaling factor (VSF) to 1 and calls to VSFMUL or VSFDIV vary this by multiplying or dividing the VSF by two respectively. The variable VVAL is used to hold an index of the current display VSF (see table 1). The routines VSFMUL and VSFDIV also contain automatic exits if they are called to yield an invalid VSF.

TABLE 1

SCALE1	SCALE2	SCALING FACTOR	VVAL
CLL RAL	CLL RAL	4	-2
* NOP	CLL RAL	2	-1
CLL RAL	NOP	2	-1
NOP	NOP	1	0
RAR	NOP	0.5	1
* NOP	RAR	0.5	1
RAR	RAR	0.25	2

* - not actually used by VSETUP.

All parameters may be varied under program control to generate different displays but in certain cases it would be better to ensure that parameters are only changed between successive calls of the display generator otherwise a specific pass of the display generator may give unspecified results, e.g. consider changing the buffer pointer and the number of points counter just before the end of a display pass which would obviously cause wrap around and certainly destroy the display until the next call. If parameters are being changed on the interrupt then it is best to arrange that a new call is issued to the routine and that the

parameters are reset as necessary. Two examples follow of parameter values to set up given displays.

(i) to generate a display of a signal average containing 400 points starting at 50 on the X-axis and with no correction factor for the data. Each point to be displayed at successive points on the display face. A baseline to be plotted at effective zero from 50 to 450 on display.

```
XAXIS,    0062      / 50 decimal
XSTEP,    0001      / successive points
BASELN,   0000      / no corrective factor
XEND,     7075      / - 451 decimal
RBIN,     7777      / 1 display point/data point
BUFFER,   DATA      / address of data buffer
NOPTS,    7160      / - 400 decimal
```

(ii) to generate a histogram display of 100 bins, each four points wide, with effective zero at the base of the display screen starting at location 50 on the X-axis, with a baseline stretching from 50 to 450 at effective zero.

```
XAXIS,    0062      / 50 decimal
XSTEP,    0001      / successive points
BASELN,   7401      / - 377 (base of screen Y-axis value)
XEND,     7075      / - 451 decimal
RBIN,     7774      / 4 point wide bin for each data point
BUFFER,   DATA
NOPTS,    7634      / - 100 decimal
```

To convert (ii) to a point display with each point being displayed four points apart XSTEP should be set to 4 and RBIN to -1.

The attached listing is heavily commented and should prove self explanatory. (the routine is shown assembled starting at loc. 0200).

```
PAGE 0001
0001
0002                /
0003                /A GENERAL PURPOSE DATA BUFFER DISPLAY ROUTINE
0004                /
0005                /A.J. SWAN    20:4:73
0006                /
0007                /FLEXIBILTY OF DISPLAY FORMAT IS ACHIEVED BY LEAVING
0008                /THE USER TO PROVIDE A SUITABLE ROUTINE TO SET
0009                /UP THE RELEVANT PARAMETERS AS DETAILED BELOW
0010                /
0011                /PARAMETERS WHICH ARE DESTROYED AND MUST BE
0012                /RESET BETWEEN EACH SUCCESSIVE CALL
0013                /
0014                /BUFFER - THE START ADDRESS OF THE DATA BUFFER
0015                /NOPTS  - MINUS THE NUMBER OF POINTS IN THE BUFFER
0016                /XAXIS  - THE INITIAL DISPLAY X-AXIS POSITION
0017                /
0018                /PARAMETERS WHICH ARE NOT DESTROYED BUT GIVE
0019                /DETAILS OF HOW THE BUFFER IS TO BE PRESENTED
0020                /ON THE DISPLAY SCREEN
0021                /
0022                /RBIN   - THE BIN WIDTH FOR EACH DATA POINT
0023                /XSTEP  - THE NUMBER OF DISPLAY POINTS BETWEEN
0024                /         EACH SUCCESSIVE INTENSIFIED POINT
0025                /BASELN - THE Y-AXIS CORRECTION FACTOR TO BE
0026                /         ADDED TO EACH DATA POINT AFTER SCALING
0027                /         TO PROVIDE EFFECTIVE ZERO CONTROL
0028                /XEND   - THE END OF THE BASELINE DRAWN ACROSS
0029                /         THE SCREEN AS DEFINED BY BASELN
0030                /SCALE1 AND SCALE2 MAY BE REPLACED BY SUITABLE
0031                /INSTRUCTIONS TO PROVIDE VERTICAL SCALING
0032                /
0033                /THE ABOVE PARAMETERS MAY EITHER BE FIXED UNDER
0034                /PROGRAM CONTROL OR IN THE SOURCE CODING
0035                /
0036 0200 0000 PLOTER, 0000
0037 0201 4264          JMS BASLIN      /PLOT BASELINE
0038 0202 1312          TAD RBIN        /RESET BINWIDTH COUNTER
0039 0203 3307          DCA BINWID
0040 0204 1710          TAD I BUFFER    /GET CURRENT DATA VALUE
0041 0205 3306          DCA CURVAL      /STORE FOR FUTURE USE
0042 0206 1313 PLOOP,   TAD XAXIS       /GET CURRENT X-AXIS POSN
0043 0207 6303          DXC DXL         /LOAD INTO DISPLAY BUFFERS
0044 0210 7200          CLA             /CLEAR VALUE
0045 0211 4243          JMS YON         /GO PLOT CURRENT COORDINATE POINT
0046 0212 1313          TAD XAXIS       /UPDATE X-AXIS VALUE
0047 0213 1314          TAD XSTEP
0048 0214 3313          DCA XAXIS
0049 0215 2307          ISZ BINWID      /DONE A BIN YET?
0050 0216 5206          JMP PLOOP       /NO SO KEEP GOING
0051 0217 2311          ISZ NOPTS       /YES SO DONE ALL DATA YET?
0052 0220 7410          SKP             /NO SO SKIP
0053 0221 5600          JMP I PLOTER    /YES SO EXIT NOW
0054 0222 2310          ISZ BUFFER      /BUMP DATA BUFFER POINTER
```

```
0055 0223 7001          IAC             /SET ACC = 1
0056 0224 1312          TAD BBIN        /ACC = 1 - BINWIDTH
0057 0225 7700          SMA CLA         /BINWID < -1?
0058 0226 5202          JMP PLOTER+2    /NO SO PLOT NEXT POINT NOW
0059                /
0060                /GET HERE IF WE ARE PLOTTING A HISTOGRAM
0061                /SO WE MUST JOIN END OF BAR REPRESENTING
0062                /THE BIN FREQUENCY BY A VERTICAL LINE TO
0063                /THE NEXT FREQUENCY
0064                /
0065 0227 1710 JOIN,    TAD I BUFFER    /GET VALUE TO BE PLOTTED NEXT
0066 0230 7041          CIA             /MAKE NEGATIVE
0067 0231 1306          TAD CURVAL      /ADD ON VALUE JUST PLOTTED
0068 0232 7450          SNA             /ARE THEY THE SAME?
0069 0233 5202          JMP PLOTER+2    /YES SO PLOT NEXT POINT NOW
0070 0234 7700          SMA CLA         /IS NEW VALUE > OLD VALUE?
0071 0235 7144          CLL CMA RAL     /SET ACC = -2
0072 0236 7001          IAC             /ACC = ACC+1 I.E. 1 OR -1
0073
0074 0237 1306          TAD CURVAL      /UPDATE VALUE OF CURVAL TO GET
0075 0240 3306          DCA CURVAL      /NEW VERTICAL LINE VALUE
0076 0241 4243          JMS YON         /DRAW LINE
0077 0242 5227          JMP JOIN        /SEE IF WE ARE THRO YET
0078                /
0079                /THIS WILL NOT DISPLAY NULL DATA
0080                /
0081 0243 0000 YON,     0000
0082 0244 1306          TAD CURVAL      /ELSE GET DATA VALUE
0083 0245 7650          SNA CLA         /NON ZERO?
0084 0246 5643          JMP I YON       /NO SO EXIT
0085 0247 1306          TAD CURVAL      /LOAD VALUE TO BE SCALED
0086 0250 7100          CLL             /ENSURE LINK IS 0
0087 0251 7510          SPA             /IS VALUE POSITIVE?
0088 0252 7020          CML             /NO SO SET LINK TO SIGN BIT
0089 0253 7000 SCALE1,  NOP             /LOCATION TO STORE FIRST SCALIN
0090 0254 7100          CLL             /AS BEFORE FOR SECOND SCALING INS
0091 0255 7510          SPA
0092 0256 7020          CML
0093 0257 7000 SCALE2,  NOP
0094 0260 1305          TAD BASELN      /ADD ON BASELINE CORRECTION
0095 0261 6317          DYC DYL DIS     /INTENSIFY POINT
0096 0262 7200          CLA             /CLEAR VALUE
0097 0263 5643          JMP I YON       /THEN EXIT
0098                /
0099                /PUTS A BASELINE ON DISPLAY AT
0100                /LEVEL DEFINED BY BASELN BETWEEN
0101                /XAXIS AND XEND
0102                /
0103 0264 0000 BASLIN,  0000
0104 0265 1313          TAD XAXIS       /SET INITIAL X-AXIS POSITION
0105 0266 3303          DCA XPOS
0106 0267 1305          TAD BASELN      /SET Y COORDINATE ONCE ONLY
0107 0270 6313          DYC DYL
0108 0271 7200          CLA
0109 0272 1303 BLOOP,   TAD XPOS        /CHECK FOR END OF LINE
```

```
0110 0273 1304          TAD XEND
0111 0274 7650          SNA CLA             /END REACHED?
0112 0275 5664          JMP I BASLIN        /YES SO EXIT
0113 0276 1303          TAD XPOS            /NO SO INTENSIFY NEXT POINT
0114 0277 6307          DXC DXL DIS
0115 0300 7200          CLA
0116 0301 2303          ISZ XPOS            /BUMP BASELINE VALUE
0117 0302 5272          JMP BLOOP           /CHECK END CONDITION
0118                /
0119                / CONSTANTS USED BY ROUTINES
0120                /
0121 0303 0000 XPOS,    0000
0122 0304 0000 XEND,    0000
0123 0305 0000 BASELN,  0000
0124 0306 0000 CURVAL,  0000
0125 0307 0000 BINWID,  0000
0126 0310 0000 BUFFER,  0000
0127 0311 0000 NOPTS,   0000
0128 0312 0000 RPIN,    0000
0129 0313 0000 XAXIS,   0000
0130 0314 0000 XSTEP,   0000
0131                /
0132                /THIS SUBROUTINE INCREASES THE VERTICAL SCALING
0133                /FACTOR BY A POWER OF TWO UP TO A MAXIMUM OF 4
0134                /
0135 0315 0000 VSFMUL,  0000
0136 0316 7305          NL0002              /SET ACC = 2
0137 0317 1352          TAD VVAL            /ADD ON CURRENT VSF INDEX
0138 0320 7650          SNA CLA             /MAX. VSF ALREADY?
0139 0321 5715          JMP I VSFMUL        /YES SO EXIT IMMEDIATELY
0140 0322 7040          CMA                 /NO SO SUBTRACT 1 FROM VSF INDEX
0141 0323 1352          TAD VVAL
0142 0324 4337          JMS VSETUP          /SET UP NECESSARY INSTRS.
0143 0325 5715          JMP I VSFMUL        /EXIT WHEN DONE
0144                /
0145                /THIS SUBROUTINE DIVIDES THE VERTICAL SCALING
0146                /FACTOR BY A POWER OF 2 DOWN TO A MINIMUM OF 0.25
0147                /
0148 0326 0000 VSFDIV,  0000
0149 0327 7344          CLA CLL CMA RAL     /SET ACC = -2
0150 0330 1352          TAD VVAL            /ADD ON CURRENT VSF INDEX
0151 0331 7650          SNA CLA             /MINIMUM VSF ALREADY?
0152 0332 5726          JMP I VSFDIV        /YES SO EXIT
0153 0333 7001          IAC                 /NO SO ADD 1 TO
0154 0334 1352          TAD VVAL            /CURRENT VSF INDEX
0155 0335 4337          JMS VSETUP          /SET UP NECESSARY INSTRS.
0156 0336 5726          JMP I VSFDIV        /THEN EXIT
0157                /
0158                /THIS SUBROUTINE SETS UP THE NECESSARY INSTRUCTIONS
0159                /IN SCALE1 AND SCALE2 FOR A GIVEN VSF INDEX ON ENTRY
0160                / IN ACC
0161                /
0162 0337 0000 VSETUP,  0000
0163 0340 3352          DCA VVAL            /STORE NEW VSF INDEX
0164 0341 1354          TAD LABVAL          /FORM ADDRESS OF FIRST INSTR.
```

```
0165 0342 1352          TAD UVAL
0166 0343 3353          DCA VSFINI      /STORE AS POINTER
0167 0344 1753          TAD I VSFIND    /LOAD FIRST INSTR.
0168 0345 3253          DCA SCALF1
0169 0346 2353          ISZ VSFINI      /PUMP INSTRUCTION POINTER
0170 0347 1753          TAD I VSFIND    /LOAD SECOND INSTR.
0171 0350 3257          DCA SCALF2
0172 0351 5737          JMP I VSETUP    /THEN EXIT
0173                /
0174                /CONSTANTS REQUIRED BY ABOVE ROUTINES
0175                /
0176 0352 0000 UVAL,    0000
0177 0353 0000 VSFIND,  0000
0178 0354 0357 LABVAL,  INSTAB
0179 0355 7104          CLL RAL
0180 0356 7104          CLL RAL
0181 0357 7000 INSTAB,  NOP
0182 0360 7000          NOP
0183 0361 7010          RAR
0184 0362 7010          RAR
0185                PAUSE
```

```
BASELN 0305    BASLIN 0264    BINWID 0307    BLOOP  0272
BUFFER 0310    CURVAL 0306    INSTAB 0357    JOIN   0227
LABVAL 0354    NOPTS  0311    PLOOP  0206    PLOTER 0200
RBIN   0312    SCALF1 0253    SCALF2 0257    VSETUP 0337
VSFDIV 0326    VSFIND 0353    VSFMUL 0315    UVAL   0352
XAXIS  0313    XEND   0304    XPOS   0303    XSTEP  0314
YON    0243
```